

ORDENADOR EDUCATIVO

PARA NOSOTROS
LA EDUCACION
DE SU HIJO
ES LO MAS
IMPORTANTE



**LA REVISTA
EDUCATIVA QUE
ESTABA ESPERANDO**

Usted tiene un ordenador. Ya ha visto las ventajas que le puede ofrecer, tanto a usted como a sus hijos. Ellos se divierten jugando, pero quisiera que sacaran más provecho de él...

Presentamos "ORDENADOR EDUCATIVO", la primera revista educativa para SPECTRUM. Contiene un CASSETTE con el cual, de una manera comprensiva, sus hijos aprenderán las materias escolares de una forma amena y diferente.

Nuestro sistema ha sido adaptado y probado por profesores y se ajusta al sistema escolar español.

De esta forma, sus hijos no sólo repasarán y estudiarán las materias escolares, sino que además se familiarizarán con la informática y su lenguaje, lo que constituye una eficaz preparación para su futuro. Esto es lo más importante para usted y nosotros.

Revista y cassette
por sólo 495 ptas.



Para envíos:

MONSER

c/ Argos, 9

28037 Madrid

Teléf. 742 72 12/96

DE VENTA EN KIOSCOS Y TIENDAS ESPECIALIZADAS



MANEJO DEL CASSETTE



O resulta nada práctico tener que teclear los programas cada vez que deseamos ejecutarlos de nuevo, sobre todo si contienen un número grande de instrucciones. La solución a estos problemas es el empleo del casete.

El Spectrum ofrece la posibilidad de utilizar un casete convencional para almacenar programas y datos. Podemos decir que para este menester sirve cualquier aparato de este tipo, pero son preferibles los equipos **mono** de bajo coste, a los **estéreo**.

Lo que también resulta deseable es que el casete disponga de cuantavuelatas, así como de un sistema de avance y retroceso rápido de la cinta, que permita la salida por el altavoz mientras se realizan estas operaciones. En todo caso, estas no son cualidades imprescindibles.

Por el contrario, es absolutamente esencial que el casete disponga de una entrada de micrófono **MIC**, y una salida para cascos **EAR**, en forma de hembras de 3,5 mm. Si el equipo no tiene este tipo de conexiones, debemos adquirir las correspondientes clavijas de adaptación en una tienda especializada. Sin embargo, esto no será necesario con la mayoría de los aparatos modernos, fundamentalmente de marcas japonesas, ya que la práctica totalidad de los mismos incorporan este tipo de conexiones.

En la parte trasera del Spectrum, entre los conectores de antena y red, existen dos conexiones marcadas **MIC** y **EAR** de 3,5 mm., que se corresponden con las entradas del mismo nombre en el casete.

Junto con el ordenador se suministra un cable doble, con dos clavijas en cada uno de sus extremos, unas de color negro y otras de color gris, que sirven para comunicar el ordenador directa-


mente con el casete; esta operación debe ser realizada entre las entradas y salidas homólogas, es decir, **MIC** con **MIC** y **EAR** con **EAR**. Como es lógico, no importa qué color de clavija se asigne a cada una de estas conexiones, pero sí es interesante que se tome una determinación para que, en adelante, utilicemos el mismo color para cada conexión en concreto. Podemos utilizar, por ejemplo, las clavijas de color gris para la conexión de **EAR** con **EAR**, y las de color negro para **MIC** con **MIC**.

Cuando deseemos almacenar el programa de la memoria en la cinta, debemos emplear las conexiones del tipo **MIC**, en nuestro ejemplo las cla-




Los usuarios que poseemos un casete, nos evitamos el trabajo de tener que teclear de nuevo los programas al encender el ordenador.

Para la grabación y lectura de programas, la utilización de equipos mono de bajo coste, suele ser preferible a la de los caros aparatos estéreo.



BASIC PARA EL CASETE



La sentencia SAVE facilita la «escritura» de programas.

El Spectrum dispone de tres palabras clave para el manejo del casete, que se corresponden con las funciones de grabación, verificación y lectura de programas, y van siempre acompañadas del nombre del programa entre comillas ("). Estas palabras son **SAVE**, en inglés salvar, **VERIFY**, en inglés verificar y **LOAD**, en inglés cargar, respectivamente.

Vamos a suponer ahora que aún permanece en la memoria nuestro último programa, y deseamos almacenarlo en la cinta con el nombre de CUADRADOS. Si no lo tenemos, porque desconectamos el ordenador al terminar de modificarlo, debemos teclearlo de nuevo. Pero esto no debe molestarnos, porque el primer paso para llegar a dominar realmente nuestro Spectrum, es adquirir práctica en el manejo del teclado. Recordemos que el programa era:

```
5 PRINT "NUMERO","CUADRADO"
10 INPUT "Numero:";N
20 LET C=N*N
30 PRINT N,C
40 GO TO 10
```

Está claro que debemos emplear la instrucción **SAVE**, que es la que se utiliza para grabar programas en cinta, pero antes debemos realizar al-

i!

Un nombre de programa está formado por una secuencia de 1 a 10 caracteres cualesquiera, siendo obligatorio especificarlo como parámetro de la sentencia **SAVE**, si bien no para **VERIFY** y **LOAD**.




La conexión entre casete y ordenador, se efectúa a través de las salidas marcadas como **MIC** y **EAR** en ambos aparatos.




No es preciso que el casete que utilizemos con el Spectrum esté dotado de una entrada para control remoto (**REMOTE**).

vijas negras, desconectando cualquiera de las grises (es indiferente la del casete o la del ordenador) para evitar problemas de interferencias en la grabación. Del mismo modo, cuando acabamos de encender el ordenador, y lo que queremos es recuperar un programa grabado anteriormente en la cinta, debemos emplear las conexiones de color gris (**EAR**), desconectando alguna de las de color negro, de forma similar al caso anterior. Es importante recordar el hecho de que, a la hora de almacenar un programa en casete, las clavijas de tipo **MIC** deben estar conectadas, mientras que alguna de las de **EAR**, o las dos, no deben estarlo.



Un casete adecuado para nuestro Spectrum puede tener las características que se reseñan en el dibujo.



BITS

La utilización con el ordenador de casetes **estéreo** no suele dar buenos resultados; por el contrario, encontraremos la máxima fiabilidad con los aparatos **mono**, de más bajo coste que los anteriores.



Aunque el Spectrum dispone de dos salidas, destinadas a las operaciones de grabación (**SAVE**) y lectura (**LOAD** y **VERIFY**), podemos realizar ambas tareas por cualquiera de ellas.



Recordemos que en el momento de realizar una grabación, no deben estar conectadas las dos salidas **EAR** de casete y ordenador.



gunas operaciones de preparación de la grabación.

La primera es avanzar la cinta hasta un sitio donde podamos grabar. Si se trata del comienzo de la misma, debemos sobrepasar la zona de cinta plástica de principio. Aunque no es obligatorio, si es muy conveniente que antes de almacenar un programa en cinta, grabemos de viva voz el nombre del mismo y lo anotemos en la carpetilla del casete junto con el número de vuelta en que se encuentra. Hecho esto, conectamos las tomas de **MIC** del ordenador y del casete a través del cable al efecto, manteniendo sin conectar una de las clavijas de **EAR** (o las dos). Una vez realizada

Existen adaptadores de clavijas que facilitan el uso de casetes que no posean conectores de 3,5 mm.

bre del programa y algunos otros datos que el ordenador necesitará más tarde para identificarlo. Esta información se denomina **HEADER** (aproximadamente se pronuncia *hider*) o **CABECERA**. Tras un cortísimo espacio de tiempo, otra vez las

Las clavijas de igual color habrán de unir las conexiones que en el Spectrum y el casete tienen el mismo nombre.




esta preparación, podemos pedir a nuestro ordenador que lleve a cabo la grabación:

SAVE "CUADRADOS"

Pulsamos **ENTER** y aparece el mensaje **Start tape, then press any key** (conecta el casete y pulsa cualquier tecla). Y es exactamente eso lo que vamos a hacer. Colocamos el casete en modo de grabación (generalmente pulsando **RECORD** y **PLAY**) y, acto seguido, pulsamos cualquier tecla del Spectrum (distinta de los **SHIFT**s).

Lo primero que vemos aparecer en el marco de la pantalla son unas bandas anchas de color rojo y azul claro, que se desplazan lentamente, durante cuatro o cinco segundos, y a continuación una ráfaga casi imperceptible, de menos de un segundo de duración, de bandas de color azul oscuro y amarillo, mucho más estrechas que las anteriores, con las que se graba en la cinta el nom-

*Con **LOAD** podemos «leer» los programas que han sido «escritos» con **SAVE**.*





bandas anchas (azules y rojas) indicativas de comienzo de bloque de datos, esta vez por espacio de unos dos segundos y, por último, nuevamente las bandas finas de color azul oscuro y amarillo, que son la representación de la grabación en cinta del bloque principal del programa.

El tiempo de grabación del último bloque de datos, varía en proporción a la longitud de cada programa en concreto. En nuestro caso será casi tan breve como el tiempo empleado para la grabación del nombre del programa; pero tratándose de programas largos, este espacio de tiempo se puede prolongar hasta algunos minutos.

El mensaje **OK** nos indica que la grabación ha terminado correctamente, al menos en cuanto al Spectrum se refiere, porque el ordenador al grabar, no detecta si existe alguna deficiencia en la superficie de la cinta que está utilizando. Para evitar el riesgo de almacenar una copia de nuestro programa que pueda ser después irrecuperable,

existe la sentencia **VERIFY**. La instrucción tiene el formato:

VERIFY "CUADROS"


Para utilizarla, rebobinamos la cinta hasta el comienzo del programa, y conectamos esta vez el cable de enlace **EAR** con **EAR**. Hecho esto, ponemos a reproducir el casete con el volumen aproximadamente a 3/4. Nos será útil recordar siempre el número de vuelta en que comienza la grabación.

Si hemos grabado el nombre del programa a viva voz, y nuestro casete nos permite dar salida al altavoz mientras rebobinamos la cinta, el comienzo se localizará muy fácilmente, ya que, frente al monótono ruido producido por los pitidos que ha grabado el Spectrum, destacará el sonido característico de una voz al ser reproducida a alta velocidad.




Para el Spectrum, la verificación consiste en leer el programa almacenado anteriormente en la cinta y compararlo con el que se encuentra en la memoria. Si los dos programas le resultan idénticos en longitud y contenido, responderá al final de la verificación con el mensaje **OK**. En caso contra-

Sistema de conexión entre casete y ordenador.




① NUNCA



② LECTURA



③ ESCRITURA





rio, se detendrá mostrando el mensaje **Tape loading error**, que nos indica que ha habido un problema en la verificación.

El proceso de verificación tiene su reflejo en el marco de la pantalla, de una forma muy similar a la que hemos comentado anteriormente para la grabación, existiendo lógicamente diferencias.


Lo primero que debemos decir es que, en este caso, el Spectrum no nos facilita el mensaje para la puesta en marcha de la cinta como la vez anterior, sino que inmediatamente después de pulsar **ENTER**, se coloca a la espera de recibir datos desde el casete. Esto lo observamos porque al conectar la clavija al casete, el Spectrum «reconoce» su conexión con él, y cambia los colores del marco de la pantalla del rojo al azul claro, de manera intermitente, para indicar esta circunstancia.

Seguidamente, aparecen las bandas anchas de color azul claro y rojo de comienzo de bloque de datos, y a continuación del **header** aparece en la pantalla el mensaje **Program: CUADRADOS**, indicativo de que el Spectrum ha identificado el nombre del programa grabado en la cinta. A continuación, vuelven a aparecer las bandas anchas hasta que el ordenador localiza el siguiente bloque de datos y se dispone a verificarlo, mostrando en el marco de la pantalla las bandas finas de colores azul oscuro y amarillo. Finalmente, aparecerá en la pantalla el mensaje de **OK**, que sirve como confirmación de que el programa quedó almacenado en la cinta correctamente.

En cualquier caso, si el proceso se interrumpe a causa de un error en la verificación, la solución es volver a grabar el programa por medio de **SAVE**, y proceder nuevamente a verificar hasta obtener los resultados apetecidos. Nadie nos libra de que, sólo de vez en cuando, tengamos algún error de verificación. Esto puede ser debido a muy diversas causas, generalmente derivadas del consiguiente deterioro de la cinta al ser borrada y vuelta a grabar en muchas ocasiones. Otras veces, el problema se debe a algún pequeño fallo en la alimentación del casete cuando éste está conectado a un adaptador de corriente alterna (sin pilas). A pesar de ello, son muchos más los problemas que pueden derivarse de realizar los procesos de grabación y recuperación de programas empleando las baterías del casete, ya que dependiendo del nivel de carga de las pilas, éste arrastra la cinta a mayor o menor velocidad. Esta diferencia, apenas perceptible para nosotros, puede ser vital para nuestro Spectrum.

De hecho, puede suceder que no se pueda cargar un programa grabado ya hace días, debido a que el estado actual de las baterías no coincide con el que se dio al efectuarse la grabación del programa. Afortunadamente, esto no suele suceder más que cuando la diferencia de velocidades es bastante notoria.

Vamos a practicar ahora con la instrucción




VERIFY comprueba que los datos «escritos» por **SAVE** coinciden con los existentes en la memoria.

LOAD, que nos permite cargar en la memoria del ordenador un programa almacenado en cinta. El formato general de la instrucción es:

LOAD "CUADRADOS"

Proceso de grabación de un programa.




Lo primero que vamos a hacer para cerciorarnos de que el programa se carga desde la cinta, es borrar la copia que tenemos ahora en la memoria por medio de la instrucción **NEW**. Pulsamos después **NEW** y **ENTER**, para obtener el mensaje de presentación del ordenador. La palabra clave **NEW** está serigrafiada en color blanco sobre la tecla de la «A», por lo que podemos acceder a ella con la sola pulsación de dicha tecla, cuando nos encontramos en el modo **K**.



Hemos de tener mucho cuidado para no intentar grabar sobre la zona plástica, que inicia y termina las cintas de casete de uso habitual.

NEW, en inglés nuevo, indica al ordenador que vamos a introducir otro programa, por lo que queremos suprimir el que reside actualmente en la memoria, y preparar nuestro Spectrum para aceptar la carga desde casete, o introducción por el teclado, de un nuevo programa.

Basándonos en las franjas que preceden a la cabecera, podemos controlar el volumen adecuado del casete.



El mensaje R Tape loading error, 0:1 aparece cuando se produce algún error durante la carga o verificación de un programa.


Para cargar el programa "CUADRADOS" desde el casete, lo primero que debemos hacer es situarnos en el lugar adecuado de la cinta. Podemos conseguirlo con la ayuda del cuentavueltas, o por medio de la orientación que nos proporciona, al rebobinar rápidamente la cinta con salida por el altavoz, nuestra voz grabada al comienzo de los programas. Una vez situados correctamente, podemos introducir la siguiente instrucción:

LOAD "CUADRADOS"

En la pantalla obtendremos exactamente los mismos resultados, que los comentados al hablar de la verificación (**VERIFY**). Primero el marco de la pantalla oscilando entre los tonos rojo y azul claro, y después las bandas anchas de estos colores, seguidas por el nombre del programa de la forma:

Program: CUADRADOS

Seguirán nuevamente las franjas anchas de color, y por último las bandas finas azules y amarillas, indicativas de que el Spectrum está leyendo el bloque principal del programa. Cuando el ordenador concluye la carga, nos envía el men-





saje de **OK** y, pulsando **ENTER**, podemos comprobar que el programa está ya copiado en la memoria, de la misma forma que antes de ejecutar la sentencia **NEW**.

OMISION DEL NOMBRE DE PROGRAMA

Los nombres de los programas están formados por series de hasta 10 caracteres de longitud, compuestas por cualquier combinación de letras mayúsculas o minúsculas, números, espacios en blanco y caracteres especiales. Tanto para la sentencia **LOAD** como para **VERIFY** puede omitirse el nombre del programa, tecleando: **LOAD ""** o **VERIFY ""**, respectivamente.

En ambos casos, el Spectrum supone que queremos cargar o verificar el programa que encuentre grabado inmediatamente a continuación en la cinta. Esto quiere decir que, si no especificamos un nombre, el ordenador no va a comprobarlo, por lo que debemos estar seguros de que el programa que se encuentra a continuación es el que deseamos. De no ser así, cargaríamos en memoria otro programa, o verificaríamos uno distinto del contenido en la memoria, con el consiguiente error.

Por el contrario, cuando introducimos un nombre de programa acompañando a la palabra clave **LOAD**, el Spectrum comprueba, al poner en marcha la cinta, los nombres de los programas que va encontrando, cargando el que coincide con el especificado. Más concretamente, el primero que comience por la secuencia de caracteres que componen el nombre del programa. Si en la cinta hemos grabado varios programas con el mismo nombre, el Spectrum cargará el primero de ellos.

①

Program:

②

Bytes:

③

Number array:

④


Character array:

Son cuatro los mensajes que pueden aparecer en la pantalla tras la lectura de una CABECERA.

SOLUCION A POSIBLES PROBLEMAS

Existen dos causas principales por las que el ordenador no conseguirá efectuar la operación de carga correctamente. La primera de ellas se refiere a problemas de volumen del casete; la segunda, a que pretendemos que el ordenador cargue un programa que realmente no lo es.

Existen varios sistemas para el almacenamiento de programas. El casete, aunque lento, es bastante fiable y de bajo coste.



i!

El volumen para las operaciones de carga y verificación de programas debe ser lo más alto posible, pero sin llegar a que el casete distorsione el sonido. Generalmente el volumen más adecuado es unos 3/4 del máximo.

*

Cuando al tratar de cargar un programa aparece en la pantalla algo distinto del mensaje **Program:**, debemos desistir de cargarlo, porque no se trata realmente de un programa sino de otra estructura de datos.

El problema más común es no poder cargar un programa debido a que el volumen del casete es inadecuado. El volumen debe colocarse siempre lo más alto posible, pero sin llegar a distorsionar, es decir, aproximadamente a 3/4 del máximo. Ya hemos dicho que, al comenzar la carga de un programa, lo primero que vemos en la pantalla son unas bandas horizontales anchas de color rojo y azul claro, que se desplazan lentamente en el marco de la misma. Estas bandas preceden al nombre del programa, y en ellas podemos advertir si el volumen es demasiado bajo, o si la velocidad de arrastre de la cinta no es uniforme. Cuando el volumen es demasiado bajo, las bandas de color son de diferentes dimensiones; la solución en ese caso es subir el volumen hasta alcanzar el nivel óptimo, en que el ancho de las bandas será similar.

Otro problema que puede presentarse es que, siendo el volumen correcto, tratemos de cargar como programa algo que no lo es. Además de programas, el Spectrum puede almacenar en la cinta, datos de código máquina, pantallas y datos numéricos o alfanuméricos.

Antes hemos visto que al cargar o verificar el programa "CUADRADOS", el Spectrum imprime el mensaje:

Program: CUADRADOS

Aunque es aconsejable realizar las cargas y verificaciones mediante el EAR del Spectrum, y las grabaciones a través del MIC, cada una de las salidas puede cumplir la función de la otra.

Además de la palabra **Program** (programa) que ya nos es familiar, puede ser que al intentar cargar una cinta comercial con **LOAD ""**, aparezca algún mensaje diferente. Eso quiere decir que la información que viene a continuación no se trata exactamente de un programa y, por lo tanto, debemos renunciar de momento e investigar sobre su contenido hasta más adelante. Los mensajes que pueden aparecer son:

Bytes:, Character array: o Numeric array:

UN EJERCICIO

Para llevar a cabo el ejercicio que proponemos, debemos borrar en primer lugar el programa que tenemos en la memoria; para ello, utilizaremos el comando **NEW**, que ya hemos estudiado anteriormente. Después vamos a introducir un nuevo programa, que puede ser cualquiera de los propuestos anteriormente; el "CUADRADOS", por ejemplo.

La primera sentencia con la que vamos a practicar es **SAVE**, que permitirá almacenar el programa en la cinta con el nombre que nosotros indiquemos. Para preparar la grabación del programa, debemos recordar colocar la cinta en un sitio donde podamos grabar y anotar el número de vuelta, así como registrar con nuestra voz el nombre que le demos al programa. Acto seguido, escribimos:


SAVE "nombre-programa"

El siguiente paso va a ser efectuar la verificación del programa. Para preparar la verificación, rebinamos la cinta hasta el número de vuelta donde sabemos que comienza el mismo, y utilizamos la palabra clave **VERIFY**, seguida del nombre que hemos dado al programa entre comillas. Como ya sabemos, un método alternativo es escribir simplemente:

VERIFY ""

No tendremos problemas, ya que como se trata del programa que acabamos de grabar, al rebinar la cinta nos posicionaremos justamente antes de su comienzo. Si todo ha marchado correctamente, se llevará a cabo la verificación, y el mensaje **OK** aparecerá en la última línea de la pantalla.

Como último paso, debemos efectuar un **NEW** para borrar el programa contenido en la memoria y, seguidamente, intentar cargarlo de nuevo con la palabra clave **LOAD**, seguida del nombre del programa (simbólicamente "nombre-programa"), o bien, mediante **LOAD ""**.





EL COLOR DE LA PANTALLA




Nuestra conciencia está la necesidad de crear bonitas pantallas. Como ya vimos anteriormente, además de un diseño adecuado de las figuras

que la integran, en esta tarea influyen además otros factores, tales como los colores empleados o el dinamismo de la pantalla en sí. A continuación daremos un pequeño repaso a varias técnicas basadas en el color de nuestro Spectrum, gracias a las cuales podremos conseguir interesantes efectos.

La gama de colores en el Spectrum es lo suficientemente variada en la mayoría de las ocasiones.

EL «BORDER» DE LA PANTALLA

Quizás podamos pensar hasta que hayamos terminado de leer este apartado, que poco se puede decir del marco de la pantalla. Nos equivocamos. Muchas sorpresas nos reserva nuestro aparato y entre ellas está la posibilidad de explotar algunas de sus sentencias BASIC, para obtener resul-




BITS

Mediante la combinación de colores en el interior de la pantalla y su marco, podemos obtener interesantes efectos ópticos. El siguiente programa es una clara demostración de este hecho.

```
10 FOR I=0 TO 7
20 PAPER 7-I: CLS:
BORDER I
30 NEXT I: GO TO
10
```



El abecedario español incorpora una letra de relativamente frecuente utilización: la Ñ. Sin embargo, esta letra no se encuentra incluida en el juego de caracteres de nuestro Spectrum. Una forma muy sencilla de obtenerla, a partir del carácter gráfico de la N, es efectuar **POKE USR "n",BIN 00111100** o bien **POKE USR "n",60**.



Cada byte del área de atributos, codifica la información sobre PAPER, INK, BRIGHT y FLASH de cada carácter de la pantalla.

tados que en un principio pueden parecer imposibles. Constatemos ahora este hecho a través de unos cuantos ejemplos.

Si aceptamos a «pies juntillas» lo que nuestro manual dice sobre el marco de la pantalla, y más concretamente sobre la sentencia **BORDER**, no cabe duda que tendremos la seguridad de no poder sacar gran partido de ese área de la pantalla. El «borde» será una zona monocolor en la que no nos es posible escribir texto o gráficos, ni tan siquiera otorgarle atributos propios de la pantalla principal, tales como intermitencia (**FLASH**), o brillo (**BRIGHT**). Intentemos ir más lejos para descubrir algunas cualidades del marco de la pantalla.

En primer lugar, hemos de saber que el cambio de color de esta zona de la pantalla, no sólo es posible realizarlo mediante la sentencia **BORDER**, sino también con un **OUT** al **PORT** adecuado, concretamente al 254. Vamos a comprobarlo: encendamos el ordenador y escribamos **BORDER 2** (y por supuesto **ENTER** al terminar, porque si no, podemos aburrirnos esperando que el ordenador haga algo). ¿Qué es lo que ha ocurrido? Sin duda el color del marco de la pantalla ha cambiado a rojo. De no ser así, o no hemos escrito **BORDER 2** o tenemos un ordenador con necesidad de acudir al «arreglólogo».

Ya podemos retornar el ordenador a su estado inicial; evidentemente utilizaremos una instrucción **BORDER 7**. A continuación, vamos a probar **OUT 254,2**. En apariencia todo ha sido igual; sin embargo, si somos un poco observadores, habremos percibido que mientras que el **OK** que sustituía a **BORDER 2** en la pantalla tenía el fondo rojo, el que ha sustituido a **OUT 254,2** no. Por tanto, la primera conclusión a la que llegamos, es que las líneas de edición toman el color del marco de la pantalla, que ha sido fijado mediante **BORDER**; sin embargo, la instrucción **OUT 254,n** no afecta a dichas líneas.

Pero aún existen más diferencias. Repite la operación anterior para retornar el color al blanco inicial. ¡Vaya sorpresa! ¿eh? No ha hecho falta más que escribir **BORDER** y el color del marco de la pantalla ha vuelto al original. Para ser más concretos, la pulsación de cualquier tecla hubiera producido este efecto. Por tanto, podemos asegurar que los efectos de **OUT 254,n** no son permanentes, si bien sí lo son los de **BORDER**. La cosa se pone interesante. Vamos a continuar por el mismo camino. Probemos ahora a ejecutar una instrucción **BORDER** con un color superior a 7 (el último admitido en la gama cromática del Spectrum). Por ejemplo: **BORDER 8**. Un error del tipo **K** (**invalid colour** = color inexistente) ha aparecido en la pantalla. Sin embargo, no sucede lo mismo si intentamos la misma estratagema con **OUT 254,8**. El color en este caso habrá pasado a negro; con **OUT 254,9** a azul, con **OUT 254,10** a rojo, y así sucesivamente. Esto quiere decir, que **OUT** admite códigos de color desde 0 hasta 255, aunque simplemente se limita a repetir la gama del Spectrum de forma cíclica.

Aún existe otra diferencia más, esta vez relacionada con el sonido. Durante el transcurso de las


La zona de memoria denominada AREA DE ATRIBUTOS, contiene toda la información referente al color de la pantalla.





pruebas que hemos realizado, no hemos percibido ninguna vibración en el altavoz al ejecutar una sentencia **OUT 254, n**. Para ello será necesario que ejecutemos varias veces consecutivas el mismo trabajo, alternando los colores del marco: **10 FOR I=0 TO 255: OUT 254,I: NEXT I**.

nítidamente varios colores en el marco; sin embargo, el movimiento de las franjas de color, no nos han dejado disfrutar plenamente del espectáculo. Una temporización adecuada nos sacará de apuros. Incluyamos un retardo mínimo en la instrucción anterior: **10 BORDER 0: BORDER**



COLORES EN EL MARCO

Vamos a abandonar ya el tema del **OUT**, pero no el del marco de la pantalla. El último experimento nos ha descubierto algo más que la emisión de sonido simultánea al color: hemos visto como el marco de la pantalla adquiría un tono indefinido y cambiante. Este es el primer paso hacia nuestro próximo objetivo: conseguir varios colores en el borde a un mismo tiempo. El sistema para conseguirlo, se basa en el «barrido» de pantalla con el que funcionan los televisores; en todo caso, no es indispensable que sepamos nada acerca de sus fundamentos y bastará con que aprendamos la técnica para conseguir este efecto.

Vamos a partir de la siguiente instrucción **10 BORDER 0: BORDER 2: BORDER 5: BORDER 6: BORDER 1: BORDER 6: BORDER 5: BORDER 2: BORDER 0: GO TO 10**

Cuando salgamos de nuestro asombro podemos pulsar **BREAK (CAPS SHIFT + SPACE)**, para detener el programa. Hemos apreciado bastante

La utilización del código de tinta 9, permite el acceso a los colores de contraste del Spectrum.

2: BORDER 5: BORDER 6: BORDER 1: BORDER 6: BORDER 5: BORDER 2: BORDER 0: PAUSE 1: GO TO 10.

¡Misión cumplida! Hemos conseguido un marco multicolor. En base a la instrucción que hemos empleado, podemos introducir pequeñas modificaciones que varíen ligeramente el resultado. Por ejemplo, en vez de utilizar 7 sentencias **BORDER**, podemos utilizar sólo 5, aunque esto producirá una inestabilidad de las franjas de color. Como habremos deducido ya, el efecto que estamos estudiando se basa en el tiempo en que el Spectrum ejecuta las instrucciones; el resultado cambiará, aunque parezca mentira, según el lugar donde se sitúe el **PAUSE**. Sin embargo, el margen que nos concede esta sentencia, para ajustar con precisión el efecto deseado, es bastante limitado. Como solución alternativa, para retardos muy pequeños, podemos utilizar los separadores dos puntos (:) entre sentencias. Por ejemplo, vamos a conseguir gracias a ellos, deslizar suavemente por la pantalla las 7 franjas que generan nuestra instrucción base:

10 BORDER 0: BORDER 2: BORDER 5: BORDER 6: BORDER 1: BORDER 6: BORDER 5: BORDER 2: BORDER 0: GO TO 10

BITS

Una forma habitual de enfatizar determinadas palabras, es subrayarlas. A continuación listamos una interesante rutina, que facilita el subrayado de mensajes en la pantalla del Spectrum.

```
10 DEF FN
Y$(X$)=X$(N)+CHR$
8+CHR$ 21+CHR$
1+" "
20 INPUT X$: FOR
N=1 TO LEN X$
30 IF X$(N)="" THEN PRINT " " : GO
TO 50
40 PRINT FN Y$(X$);
50 NEXT N
```



Son muchas las sorpresas que nos reserva la ROM de nuestro Spectrum. Por ejemplo, seguro que muchos de nosotros desconocíamos que la acción simultánea sobre las teclas **CAPS SHIFT**, **SYMBOL SHIFT** y **Z**, fuerza la detención del programa, de la misma manera que lo hace **BREAK (CAPS SHIFT+SPACE)**.



BITS

Acudiendo a la ROM podemos solucionar muchos problemas que implican el manejo del código máquina. Tal es el caso de la subrutina de **scroll**, que se encuentra localizada en la dirección decimal 3190. Efectuando **RANDOMIZE USR 3190**, conseguiremos realizar un desplazamiento de la pantalla. Para comprobarlo introducimos el siguiente programa.

```
10 FOR I=0 TO 25
20 PRINT AT
21,0;"LINEA No.:";I
30 PAUSE 30:RANDOMIZE USR 3190:
NEXT I
```


Para ello, hemos introducido 27 separadores entre **BORDER 0** y **GO TO 10**. Si nos dedicamos a hacer pruebas, comprobaremos que hasta la inclusión de un separador más o menos, influye en el resultado final. Utilizando código máquina, se dispone de una temporización extraordinariamente precisa, gracias a lo cual se pueden conseguir gran cantidad de franjas estables.

Para finalizar con el tema del marco de la pantalla, hemos de recordar que no es sólo importante cambiar su color, sino que también cumple una misión fundamental en la estética de la presentación, sin necesidad de utilizar técnicas especiales. Tengamos siempre presente que un color de marco igual al de pantalla interior, produce un efecto óptico que aumenta aparentemente las dimensiones del conjunto «marco-interior». Por el contrario, utilizar colores distintos para las dos zonas de la pantalla, puede ayudarnos a crear una delimitación del «área de juego o acción», sin necesidad de recurrir al trazado de líneas u otros métodos.


LA PALETA DEL PINTOR

Todos los pintores tienen su paleta de colores y nosotros, los diseñadores de pantallas, no vamos a ser menos. Para nuestra desgracia, los artistas del pincel electrónico no disponemos en el Spec-


①



②




La representación de dos líneas con distinto color de primer término, sólo es posible cuando están controladas por diferentes bytes del área de atributos, es decir, cuando pertenecen a dos caracteres diferentes.



trum de una gran variedad de tonalidades; sin embargo, esto no debe ser disculpa para no combinar adecuadamente aquellos colores que sí podemos emplear.

Para empezar, nunca debemos despreciar las grandes posibilidades que aporta el modo **BRIGHT**, ya que, en muchas ocasiones, las diferencias entre los tonos con y sin brillo son ostensibles. Veamos un ejemplo: encendamos el ordenador y ejecutemos la instrucción **BRIGHT 1:CLS**. Aparentemente, la pantalla ha continuado siendo blanca y el marco ha adquirido un color gris claro. Este efecto óptico se debe a la diferencia de brillo entre ambas zonas, y por supuesto lo podemos explotar a nuestra conveniencia. Lo mismo sucede cuando combinamos ambos brillos de amarillo, gracias a lo cual el ama-




Es posible, de una manera muy sencilla, obtener marcos de pantalla multicolor.

rillo que normalmente utilizamos adquiere un tono verdoso.

Efectos similares se pueden obtener con todos los colores, aunque según vamos adentrándonos en la gama, menor es la diferencia entre ambas tonalidades de cada color, hasta llegar al negro, donde la diferencia se hace prácticamente imperceptible.

Una vez conscientes de que nuestra paleta no se

Una pequeña diferencia en el tiempo de ejecución de las instrucciones generadoras del marco multicolor, produce un desplazamiento de las franjas de color.



reduce a 8 tonos, sino a 16. Hemos de manejar con cuidado los colores de fondo y primer término, es decir, **PAPER** e **INK**, respectivamente. Nunca debemos utilizar colores de una excesiva proximidad en la gama, pues a no ser que estemos tratando grandes superficies, la visibilidad del primer término sobre el fondo se hace bastante escasa. Este desagradable efecto, queda patente en la escritura de mensajes. He aquí un ejemplo clarísimo de lo que estamos diciendo:

```
PRINT PAPER 7; INK 6;"MENSAJE"
PRINT PAPER 7; INK 0;"MENSAJE"
```

No debe existir duda alguna entre cual de los dos colores ha conseguido una mejor visibilidad en el mensaje escrito. Para evitar este tipo de «desastres visuales» de una manera cómoda, podemos dejar a juicio del ordenador el tono en que debe escribir. Para ello utilizaremos un color especial, fuera de la gama habitual, el **9**. Este código es el que se conoce como **TINTA DE CONTRASTE**. Gracias a él, lo que se escriba en la pantalla, cualquiera que sea su color de fondo, será perfectamente legible.

Para los cuatro primeros colores de la gama (del negro al magenta), el Spectrum elige el blanco como tinta de contraste. Para los restantes (del verde al blanco), la tinta elegida es la negra. Este es el mismo sistema que el Spectrum sigue para escribir en la línea de edición. Con la siguiente rutina veremos como el ordenador ajusta el color de la tinta mediante el código **9**.

```
10 FOR I=0 TO 7
20 BORDER I: PAPER I: CLS
30 PRINT INK 9;"MENSAJE"
40 PAUSE 100
50 NEXT I
```

De la misma manera que **INK 9** ajusta el color de primer término, como contraste con el de fondo, **PAPER 9**, contrasta éste con el de tinta. Pese a todo lo dicho, en ocasiones no es tarea fácil compatibilizar varios colores en puntos (*pixels*) muy cercanos, como sería nuestro deseo. Esto sucede frecuentemente al trazar líneas de diferente color con la sentencia **DRAW**.

Hay que tener en cuenta que a cada posición de carácter de la pantalla, le corresponde un solo byte de la zona de atributos y, por tanto, únicamente se puede combinar un color de fondo y uno de primer término en un mismo carácter. De igual modo, cualidades como el brillo afectan simultáneamente tanto a papel como a tinta. Así pues, la única solución para escribir dos puntos consecutivos con distinto tono de primer término, es situarlos en caracteres diferentes, con el fin de que sean controlados con bytes distintos del área de atributos.



BITS

He aquí un programa con el que podremos dejar admirados a nuestros amigos. Gracias a él, generaremos espectaculares gráficos de alta resolución, jempleando sólo tres instrucciones!

```
10 PLOT 90,80
20 INPUT N
30 DRAW 60,60,N:
PAUSE 100: RUN
```

Dependiendo del valor que asignemos a la variable **N**, mediante el **INPUT** de la línea 20, obtendremos diferentes resultados. Podemos empezar probando con los siguientes valores: 7689, 8462, 3572, 1010, 5387, 3546, 2547, 4189 y 8888.



El efecto de parpadeo que produce la sentencia **FLASH** del Spectrum, carece en ocasiones de la estética deseable para la impresión de mensajes. Un método alternativo al **FLASH** del sistema, puede ser escribir y borrar el mensaje cíclicamente. Este efecto lo podíamos conseguir de una manera muy simple, utilizando repetidamente la sentencia **OVER**.

```
10 PRINT AT
11,6:OVER 1;"MENSAJE PARPADEANTE": PAUSE 25: GO
TO 10
```




ADIVINADOR

!

Para grabar el programa, utilizaremos la instrucción SAVE "ADIVINADOR" LINE 10. Si suprimiésemos la sentencia LINE 10, el programa no se autoejecutaría al finalizar la carga del mismo.

*

El vocabulario especificado en las dos primeras DATAs del listado, se refiere por entero a vocablos informáticos. Podemos variar el contenido de dichas DATAs, teniendo en cuenta que el número de palabras que hay en cada una de ellas ha de ser constante.

*

Las subrutinas de programa que son accedidas desde otras subrutinas, se denominan subrutinas anidadas.

*


El puzzle de letras se desordena en la zona de programa 380-430, apoyándose en las cadenas A\$, P\$ y O\$.



El programa que a continuación presentamos, se encuentra dentro del grupo de los llamados juegos de lógica; el problema que plantea es el siguiente: el ordenador nos reta a que adivinemos un número que él mismo ha «pensado». Dicho número está comprendido entre 1 y un valor determinado (número tope), prefijado por nosotros mismos. A partir de ese momento, comienza el juego, debiendo hacer un esfuerzo por adivinar la cifra propuesta en el menor número de intentos posible.

Pero no creamos que va a ser tan fácil. Por cada cinco intentos consecutivos sin conseguir desvelar el número secreto, tendremos que superar una prueba propuesta por el retorcido «cerebro» de nuestro Spectrum: debemos reconstruir una palabra previamente descompuesta por el ordenador, y disponemos para ello de sólo cinco oportunidades.

En caso de no ordenar correctamente la palabra desmantelada, en el límite fijado, el Spectrum elegirá otro nuevo «número incógnita» dentro del intervalo inicial predefinido, resultando por tanto



Por cada cinco «ensayos adivinatorios» errados, el programa propone la recomposición de un puzzle de letras, con el tema informático de fondo.

Al empezar, el ordenador nos mostrará en la pantalla los valores que definen el intervalo en el que se encuentra el «número incógnita». Sólo nos resta adivinar el número elegido por el Spectrum. Para ello, iremos tanteando el intervalo en sucesivos intentos.

Por cada número introducido, aparecerá un mensaje que dependerá de la comparación lógica entre el valor a adivinar y el dato numérico anteriormente tecleado. Así pues, sólo pueden darse tres casos, y por tanto tres mensajes diferentes:

1. Número introducido = Número incógnita
MENSAJE: ACERTASTE
2. Número introducido > Número incógnita
MENSAJE: INTRODUCE UN NUMERO MENOR
3. Número introducido < Número incógnita
MENSAJE: INTRODUCE UN NUMERO MAYOR

baldíos nuestros anteriores intentos de adivinar el número. Por el contrario, si logramos dar con la palabra desordenada dentro de las cinco oportunidades que el ordenador nos concede, podremos continuar el juego donde lo habíamos dejado.

Una vez adivinado el número, podremos optar por el cambio del límite superior que define el tope del intervalo, en el que nuestro Spectrum escoge el valor numérico que hemos de descubrir.


EL PROGRAMA


La arquitectura del programa está compuesta por una serie de subrutinas, que son llamadas bien



desde el programa principal, o bien desde otra subrutina (**subrutinas anidadas**).

Por lo que respecta a las variables, cabe observar que se reducen estrictamente a las imprescindibles para la adecuada elaboración y ejecución del programa. Las variables numéricas FL, PAP, F y C, contienen una información muy útil, que propicia una presentación agradable de los mensajes en la pantalla.

Finalmente, hay que destacar que el banco de datos, con el que el ordenador trabaja para proponer y desordenar palabras, reside en las dos primeras líneas **DATA**, y está compuesto enteramente por vocablos relacionados con la informática, de manera que cuanto mayor sea nuestra cultura en este campo del saber, más probabilidades tendremos de adivinar el número en pocos intentos. 





PROGRAMA



```
10 REM *****
20 REM ***** J.M.MAYORAL *****
30 REM *****
40 PAPER 0: BORDER 0: CLS
50 POKE 23658,8
60 LET MU=1
70 GO TO 890
80 REM SBR.ELIGE NUMERO
90 INPUT "NUMERO TOPE = ";NT: GO SUB 1630
100 LET NUM=INT (RND*NT)+1
110 RETURN
120 REM
130 REM SBR.ESCRIT.HORIZ.
140 FOR N=1 TO LEN A$
150 PRINT PAPER PAP; FLASH FL; AT F,C+N;A$(N)
160 NEXT N
170 RETURN
180 REM SBR.BORRADO
190 FOR N=1 TO 30
200 PRINT PAPER 0; AT F,N; " "
210 NEXT N
220 RETURN
230 REM BORRADO PANTALLA
240 FOR F=1 TO 20
250 PRINT PAPER 0; AT F,1; " "
260 NEXT F
270 RETURN
280 REM SBR.MULTIPLO DE 5
290 IF CONT=MU*5 THEN LET MU=MU+1: GO TO 450
300 RETURN
310 DATA "ACUMULADOR","BYTE","OCTETO","BIT","COMPILA
DOR","FIRMWARE","BASIC","SOFTWARE","HARDWARE","CHIP",
"SENTENCIA","SCROLLING","BAUDIO","BUG","RAM","ROM","C
ODIGO","INTERFACE","BUCLE","HIT","CURSOR","CPU","INFO
RMATICA","INICIALIZACION","ROUTINA","SUBPROGRAMA"
320 DATA "IMPRESORA","JOYSTICK","EDITAR","INTERACION",
"MONITOR","MODEM","TERMINAL","VARIABLE","CAMPO","REG
ISTRO","MENU","KBYTE","PROGRAMA","MATRIZ","TABLA","EN
SAMBLADOR","DESENSAMBLADOR","SPRITE","WINCHESTER","HE
URISTICO","DOWNLOAD","ALGORITMO","PSEUDOCODIGO","ASCI
I","BOOTING","BOOTSTRAPPING","CARACTER","BUS","PORT",
"CANAL","PILA","PERIFERICO","PIXEL","PROMPT"
330 RESTORE 310
340 FOR N=1 TO INT (RND*60)+1
350 READ P$
360 NEXT N
370 DIM O$(LEN P$)
380 FOR N=1 TO LEN P$
390 LET L=INT (RND*LEN P$)+1
400 IF O$(L)<>" " THEN GO TO 390
410 BEEP .1,20: LET O$(L)=P$(N)
420 IF O$=P$ THEN BEEP 1,20: GO TO 310
430 NEXT N
440 RETURN
450 REM SBR.PASAT
460 GO SUB 230
470 LET A$="5 INTENTOS SEGUIDOS FALLADOS"
480 LET FL=1: LET PAP=2: LET F=10: LET C=1
490 GO SUB 140
500 LET A$="RECOMPON LA PALABRA..."
510 LET FL=0: LET PAP=2: LET C=4: LET F=12
520 GO SUB 140
530 GO SUB 310
540 LET A$=O$
550 LET FL=0: LET F=14: LET PAP=0: LET C=INT ((32-LE
N P$)/2)-2
560 GO SUB 140
570 PRINT AT 19,6: "OPORTUNIDADES = ";5
580 FOR X=4 TO 0 STEP -1
590 INPUT "LA PALABRA ES.. "; LINE I$
600 BEEP .1,30
610 IF I$=P$ THEN GO TO 810
620 LET A$="PALABRA INCORRECTA"
630 LET PAP=2: LET FL=1: LET F=17: LET C=5
640 GO SUB 140: BEEP 1,-10: GO SUB 180
650 PRINT AT 19,6: "OPORTUNIDADES = ";X
660 NEXT X
670 PRINT PAPER 0; AT 19,3; " "
680 LET A$=" LA PALABRA ES "
690 LET PAP=6: LET F=17: LET C=2: LET FL=0
700 GO SUB 140
710 LET A$=P$
720 LET PAP=0: LET F=19: LET C=INT ((32-LEN P$)/2)-2
730 GO SUB 140
740 FOR G=1 TO 10: BEEP .1,40: BEEP .1,35: BEEP .1,3
0: NEXT G
750 PAUSE 100
760 GO SUB 230
770 LET A$="ELIJO OTRO NUMERO"
780 LET F=10: LET C=6: LET FL=0
790 GO SUB 140: PAUSE 50: GO SUB 180
```

```
800 LET CONT=0: LET MU=1: LET V=0: GO SUB 100: GO TO
910
810 LET V=1
820 GO SUB 230
830 LET A$=" CORRECTO "
840 LET PAP=6: LET FL=1: LET F=10: LET C=9
850 GO SUB 140
860 BEEP .5,10: BEEP .5,20
870 GO SUB 180
880 RETURN
890 REM PROGRAMA PRINCIPAL
900 GO SUB 80
910 PRINT INK 6; PAPER 5; AT 21,0; " "
920 PRINT INK 6; PAPER 5; AT 0,0; " "
930 INK 9
940 FOR F=1 TO 20
950 PRINT PAPER 5; INK 6; AT F,0; " "; AT F,31; " "
960 NEXT F
970 PRINT AT 7,6: "EL NUMERO QUE DEBES"
980 PRINT AT 9,8: "ADIVINARME ESTA"
990 PRINT AT 11,6: "COMPRENDIDO ENTRE..."
1000 PRINT PAPER 2; AT 13,11; "1"; PAPER 0; " Y "; PA
PER 2; NT
1010 GO SUB 1560
1020 FOR N=1 TO 250
1030 NEXT N
1040 LET A$="PULSA UNA TECLA PARA EMPEZAR"
1050 LET PAP=0: LET FL=0: LET F=19: LET C=1
1060 GO SUB 140
1070 IF INKEY$="" THEN GO TO 1070
1080 BEEP .3,50
1090 GO SUB 230
1100 LET CONT=0
1110 LET A$="INTRODUCE NUMERO"
1120 LET PAP=0: LET FL=1: LET F=19: LET C=7
1130 GO SUB 140
1140 INPUT NC
1150 PRINT PAPER 0; AT 5,1; " "
1160 LET K$=STR$ NC
1170 LET F=19: GO SUB 180
1180 LET F=10: GO SUB 180
1190 LET CONT=CONT+1
1200 IF NUM>NC THEN GO TO 1230
1210 IF NUM<NC THEN GO TO 1300
1220 IF NUM=NC THEN GO TO 1370
1230 REM NUM>NC
1240 GO SUB 280
1250 LET A$="INTRODUCE UN NUMERO MAYOR"
1260 LET PAP=6: LET FL=0: LET F=10: LET C=2
1270 GO SUB 140
1280 PRINT AT 5,INT ((32-LEN K$)/2);NC
1290 GO TO 1110
1300 REM NUM<NC
1310 GO SUB 280
1320 LET A$="INTRODUCE UN NUMERO MENOR"
1330 LET PAP=2: LET FL=0: LET F=10: LET C=2
1340 GO SUB 140
1350 PRINT AT 5,INT ((32-LEN K$)/2);NC
1360 GO TO 1110
1370 REM NUM=NC
1380 GO SUB 230
1390 LET A$=" A C E R T A S T E "
1400 LET PAP=5: LET FL=1: LET F=10: LET C=5
1410 GO SUB 140
1420 LET A$="LO ADIVINASTE EN "
1430 LET PAP=0: LET FL=0: LET F=19: LET C=1
1440 GO SUB 140
1450 PRINT PAPER 6; AT 19,19; CONT; PAPER 0; " INTENTOS
1460 FOR N=-20 TO 40
1470 BEEP .01,N
1480 BORDER 5: BORDER 2: BORDER 1: BORDER 6
1490 NEXT N
1500 BORDER 0
1510 GO SUB 230
1520 PRINT AT 19,1: "DESEAS CAMBIAR EL NUMERO TOPE?"
1530 IF INKEY$="S" THEN RUN
1540 IF INKEY$="N" THEN GO SUB 230: GO SUB 100: LET
MU=1: LET CONT=0: GO TO 910
1550 GO TO 1530
1560 RESTORE 1570
1570 DATA 74,46,32,77,46,32,77,65,89,79,82,65,76
1580 FOR N=1 TO 13
1590 READ A
1600 PRINT INK 5; AT 2,N+8; CHR$ A
1610 NEXT N
1620 RETURN
1630 FOR N=1 TO 10
1640 LET L=RND*ABS (SQR N*NT)
1650 NEXT N
1660 RETURN
```